

SYMple Clock

A. M. MacKay
Owen Sound, Ontario

There are lots of clock programs around, but this one is a little different, and a lot more useful than most. It is written for a SYM-1 with 4K memory, but will work with a bare board. It is similar in some ways to the one in Issue 1 of Compute II, but doesn't require Basic or a CRT.

While most clock programs using the LED readouts won't do anything else while the clock is running, this one sits at the top of your memory, out of the way of most programs, and quietly ticks away while you and your SYM do other and better things. But when you want the time, either visually or for use by a program, just call it and there it is.

Since a program such as this will likely be stored on tape and loaded when required over the years, no page zero slots are used so you don't have to remember or keep track of them.

The clock runs in 24 hour format. If you prefer 12, change location 0FAB to 12. Then to start the clock, enter the hours, minutes and seconds, with fifteen or twenty seconds lead time, into locations 0FFD, 0FFE and 0FFF respectively. Then enter "GO F3F" and at the exact second hit "CR" and presto! Nothing happens! Ah, but it does. Your clock is running, quietly minding it's own business, eagerly awaiting your summons. Now, to see the time, all you do is hit "SHIFT CALC CR" or "SHIFT 0 CR" or any other "UNRECOGNIZED" command. The time will be displayed for a few seconds, then the readouts will be blanked except for a row of dots. As soon as you see the dots, you can go back to

whatever you were doing with your SYM. If you want the time displayed during, and as part of, a program, just use "JSR B9 0F" and there it is.

Most importantly, though, if you want the time for controlling purposes, just call it at 0FFD, 0FFE and 0FFF with your program. It can, at the proper time, sound an alarm, turn off the lights, turn on your lights, and/or whatever makes you happy.

If you don't want clock time, but just the time since your SYM was turned on (actually since the clock started), don't enter anything in 0FFD-F. Just hit "GO F3F CR" and the clock will automatically start at 00 hours 00 minutes 00 seconds.

If your SYM is new, this is a good chance to experiment, changing things to suit your purposes. For example, try moving "DLY" from line 1090 to line 1010 (change "EB" to "D4" at location 0FEA). Your SYM now looks like a cheap digital clock. Now try changing "0A" at location 0FD2 to "1A", then move lines 1070 and 1080 to a new location between lines 1000 and 1010. Do you prefer the display this way?

The theory of operation is similar to that given for my clock article in Compute II no. 1, which required Basic and a CRT. However, the program is somewhat different because the clock in that article kept time in hex, while this one keeps time in decimal. To work in decimal with "SED" you must use ADC or SBC. "INC" just doesn't work.

Your SYM-1 is very powerful by itself, and is the basis for an extremely complex and powerful system. To get the most from it, I urge you to join the SYM-1 Users' Group, P.O. Box 315, Chico, CA 95927. And, of course, subscribe to and keep reading COMPUTE!

```

0005      .OS
0010 ;      *****
0020 ;      *****
0030 ;      ***
0040 ;      *** > > > SYMPLE  CLOCK < < < ***
0050 ;      ***
0060 ;      *** BY A. M. MACKAY ***
0070 ;      *** CO-ORDINATOR, SURVEY DEPT. ***
0080 ;      *** GEORGIAN COLLEGE A.A.T. ***
0090 ;      *** 1150 EIGHTH STREET EAST ***
0100 ;      *** OWEN SOUND, ONTARIO CANADA ***
0110 ;      *** N4K 5R4 JUNE 18, 1980 ***
0120 ;      ***
0130 ;      *****
0140 ;      *****
0150 ;
0160 ;
0170      .BA $0F3F
0180 ;
0190 ;      * * * DEFINITION OF LABELS * * *
0200 ;
0210 OUTBYT      .DE $82FA
0220 SCAND       .DE $8906

```

```

0230 ACCESS      .DE $8B86
0240 IFR2        .DE $A405      ;FLAG FOR DISPLAY TIMER
0250 DISBUF      .DE $A640
0260 URCVEC      .DE $A66D
0270 IRQVEC      .DE $A67E
0280 CLRINT      .DE $AC04
0290 TICH        .DE $AC05
0300 TILL        .DE $AC06
0310 ACR         .DE $AC0B
0320 IFR         .DE $AC0D
0330 IER         .DE $AC0E
0340 ;
0350 ;          * * * INITIATE TIMER * * *
0360 ;
0F3F- 20 86 8B 0370 START      JSR ACCESS      ;UNWRITE PROTECT SYS RAM
0F42- A9 71      0380          LDA #L,CLOCK ;SET IRQ
0F44- 8D 7E A6 0390          STA IRQVEC      ; VECTOR
0F47- A9 0F      0400          LDA #H,CLOCK ; TO
0F49- 8D 7F A6 0410          STA IRQVEC+1 : "CLOCK"
0F4C- A9 B9      0420          LDA #L,TIME ;SET UNRECOGNIZED
0F4E- 8D 6D A6 0430          STA URCVEC      ; COMMAND VECTOR
0F51- A9 0F      0440          LDA #H,TIME ; TO
0F53- 8D 6E A6 0450          STA URCVEC+1 : "TIME"
0F56- A9 C0      0460          LDA #$C0 ;SET BITS 6 & 7
0F58- 8D 0B AC 0470          STA ACR ; FOR FREE RUNNING MODE
0F5B- 8D 0E AC 0480          STA IER ; AND T1 INTERRUPT ENABLE
0F5E- AD 0D AC 0490          LDA IFR ;CLEAR T1 FLAG BIT 6 BUT
0F61- 29 BF      0500          AND #$BF ; DON'T DISTURB OTHER
0F63- 8D 0D AC 0510          STA IFR ; IFR BITS
0F66- A9 50      0520          LDA #$50 ;SET
0F68- 8D 06 AC 0530          STA TILL ; TIMER
0F6B- A9 C3      0540          LDA #$C3 ; FOR 1/20 SEC AND
0F6D- 8D 05 AC 0550          STA TICH ; START TIMER
0F70- 60      0560          RTS
0570 ;
0580 ;          * * * INTERRUPT SERVICE ROUTINE * * *
0590 ;
0F71- 48      0600 CLOCK      PHA ; ;SAVE ACCUMULATOR
0F72- F8      0610          SED ; ;TIME IS IN DECIMAL MODE
0F73- CE FC 0F 0620          DEC COUNT ;SEE IF 1 SEC HAS PASSED
0F76- D0 3B      0630          BNE EXIT ;IF NO, EXIT
0F78- A9 14      0640          LDA #20 ;IF YES,
0F7A- 8D FC 0F 0650          STA COUNT ; RESTORE COUNT
0F7D- 18      0660          CLC ; AND
0F7E- A9 01      0670          LDA #01 ; ADD 1
0F80- 6D FF 0F 0680          ADC SECS ; TO
0F83- 8D FF 0F 0690          STA SECS ; SECS
0F86- C9 60      0700          CMP #$60 ;SEE IF 60 SECS HAS PASSED
0F88- D0 29      0710          BNE EXIT ;IF NO, EXIT
0F8A- A9 00      0720          LDA #00 ;IF YES, RESET
0F8C- 8D FF 0F 0730          STA SECS ; SECS TO ZERO
0F8F- 18      0740          CLC ; AND
0F90- A9 01      0750          LDA #01 ; ADD
0F92- 6D FE 0F 0760          ADC MINS ; ONE TO
0F95- 8D FE 0F 0770          STA MINS ; MINS
0F98- C9 60      0780          CMP #$60 ;SEE IF 60 MINS HAS PASSED
0F9A- D0 17      0790          BNE EXIT ;IF NO, EXIT
0F9C- A9 00      0800          LDA #00 ;IF YES, RESET
0F9E- 8D FE 0F 0810          STA MINS ; MINS TO ZERO
0FA1- 18      0820          CLC ; AND

```

```

0FA2- A9 01      0830      LDA #01      ;      ADD
0FA4- 6D FD 0F   0840      ADC HOUR      ;      ONE TO
0FA7- 8D FD 0F   0850      STA HOUR      ;      HOUR
0FAA- C9 24      0860      CMP #24      ;SEE IF 24 HOURS HAS PASSED
0FAC- D0 05      0870      BNE EXIT      ;IF NO, EXIT
0FAE- A9 00      0880      LDA #00      ;IF YES, RESET
0FB0- 8D FD 0F   0890      STA HOUR      ; HOUR TO ZERO
0FB3- AD 04 AC   0900 EXIT    LDA CLRINT   ;ENABLE TIMER INTERRUPT
0FB6- D8          0910      CLD          ;BACK TO HEX
0FB7- 68          0920      PLA          ;RESTORE ACCUMULATOR
0FB8- 40          0930      RTI          ;
                                0940 ;
                                0950 ;      * * * DISPLAY ROUTINE * * *
                                0960 ;
0FB9- 20 86 8B   0970 TIME    JSR ACCESS   ;UNWRITE PROTECT SYS RAM
0FBC- 48          0980      PHA          ;SAVE ACCUMULATOR
0FBD- 8A          0990      TXA          ; AND
0FBE- 48          1000      PHA          ; X-REGISTER
0FBF- AD FD 0F   1010      LDA HOUR      ;PUT
0FC2- 20 FA 82   1020      JSR OUTBYT
0FC5- AD FE 0F   1030      LDA MINS      ; TIME ON
0FC8- 20 FA 82   1040      JSR OUTBYT
0FCB- AD FF 0F   1050      LDA SECS
0FCE- 20 FA 82   1060      JSR OUTBYT   ; DISPLAY
0FD1- A9 0A      1070      LDA #0A      ;SET NUMBER OF
0FD3- 8D FB 0F   1080      STA CNT1     ; TIMEOUTS FOR DISPLAY
0FD6- A9 FF      1090 DLY      LDA #FFF   ;SET LENGTH OF
0FD8- 8D 1F A4   1100      STA $A41F    ; TIMEOUT
0FDB- 20 06 89   1110 DISPL   JSR SCAND   ;LIGHT LEDS
0FDE- AD FA 0F   1120      LDA MASK     ;CHECK TIMER
0FE1- 2C 05 A4   1130      BIT IFR2     ; IRQ
0FE4- 10 F5      1140      BPL DISPL    ;IF NO IRQ REPEAT
0FE6- CE FB 0F   1150      DEC CNT1     ;ELSE START AGAIN
0FE9- 10 EB      1160      BPL DLY      ;FINISHED?
0FEB- A2 05      1170      LDX #05      ;CLEAR
0FED- AD FA 0F   1180 CLR      LDA MASK
0FF0- 9D 40 A6   1190      STA DISBUF,X : DISPLAY
0FF3- CA          1200      DEX
0FF4- 10 F7      1210      BPL CLR
0FF6- 68          1220      PLA          ;RESTORE
0FF7- AA          1230      TAX          ; X-REGISTER AND
0FF8- 68          1240      PLA          ; ACCUMULATOR
0FF9- 60          1250      RTS
                                1260 ;
                                1270 ;      * * * STORAGE DEFINITIONS * * *
                                1280 ;
0FFA- 80          1290 MASK     .BY %10000000 ;BIT 7 ONLY
0FFB-          1300 CNT1      .DS 1         ;PROVIDE SPACE FOR CNT1
0FFC- 14          1310 COUNT   .BY 20       ;SET COUNT TO 20
0FFD- 00          1320 HOUR     .BY 00       ;START TIME AT 00 HOURS
0FFE- 00          1330 MINS     .BY 00       ; 00 MINUTES
0FFF- 00          1340 SECS     .BY 00       ; 00 SECONDS
                                1350      .EN

```

LABEL FILE: [/ = EXTERNAL]

```

/OUTBYT=82FA      /SCAND=6906      /ACCESS=9886
/IFR2=A405        /DISBUF=A640      /JACVEC=A660
/IRQVEC=A67E      /CLRINT=AC04      /TICK=AC05
/TILL=AC36        /ACB=AC08        /IFR=AC0D
/IER=AC0E         START=0F3F      CLOCK=0F71
EXIT=0FD3         TIME=0FB9        DLY=0FD6
DISPL=0FDB        CLR=0FED        MASK=0FFA
CNT1=0FE2         COUNT=0FFC       HOUR=0FFD
MINS=0FFE         SECS=0FFF
/0000,1000,1000
>

```